

**Comparison of the Lake Powell Pipeline Financial Model  
previously endorsed by  
some Utah Academic Economists  
with the Lake Powell Pipeline Financial Model of the  
Washington County Water Conservancy District's  
Consulting Firm**

Gabriel A. Lozada\*  
Gail Blattenberger

Department of Economics, University of Utah,  
Salt Lake City, UT 84112

September 2016

---

\*Corresponding author: [lozada@economics.utah.edu](mailto:lozada@economics.utah.edu); 801-581-7650.

Section I of this report describes the Lake Powell Pipeline financial model of the Washington County Water Conservancy District's consultant, Jeremy Aguero, and his firm, Applied Analysis of Las Vegas, Nevada. This model is contained in an Excel spreadsheet which was released to the public in June 2016 in response to a request under Utah's Governmental Records and Management Act ("GRAMA"). A copy of this spreadsheet can now be downloaded from [http://economics.utah.edu/lozada/Research/WCWCD Model 11 November 2013 Meeting \(Working Draft\).xlsx](http://economics.utah.edu/lozada/Research/WCWCD Model 11 November 2013 Meeting (Working Draft).xlsx). Appendices A–F of this report support Section I and can be found at the end of this document.

For comparison, Section II of this report reviews the report endorsed by some Utah academic economists in 2015. The letter by the academic economists is at [http://economics.utah.edu/lozada/Research/2015\\_LPP\\_Economic\\_Analysis.pdf](http://economics.utah.edu/lozada/Research/2015_LPP_Economic_Analysis.pdf) and the model behind it is at [http://economics.utah.edu/lozada/Research/2015\\_LPP\\_FINAL\\_Repayment\\_Analysis10.xlsm](http://economics.utah.edu/lozada/Research/2015_LPP_FINAL_Repayment_Analysis10.xlsm).<sup>1</sup> This will be referred to as "the economists' model."

"Washington County Water Conservancy District" will be abbreviated WCWCD, and the model of its consultant will for short be referred to as the District's model or the WCWCD model. The Lake Powell Pipeline will be abbreviated LPP. In the electronic version of this document, the URLs are "clickable," i.e., they are hyperlinks.

### **Section I: The WCWCD Model**

The main calculations of the District's model are in the sheet named "LPP Act Financing Plan," or LAFP. It starts with Row 6, "Population," which is the same as in the Economists' Model; that model in turn obtained it from the Governor's Office of Planning and Budget 2012 Baseline Population Projections. Row 7 gives "Incremental ERUs," where ERU is "equivalent residential unit"; Row 7 is obtained by calculating population growth and dividing by 'Underlying Assumptions'!\$E10, the "number of people per ERU," assumed to be 2.94. (The GOPB 2012 analysis takes that to be 3.16.) Water demand, Row 9, is given in "gallons per capita per day," "GPCD." This is assumed to fall to 250 GPCD by 2050 ('Conservation Input'!\$I\$22) and then remain constant. Routine calculations then lead to a projection of "incremental share of Lake Powell [pipeline] Capacity Utilized," Row 23.

In the WCWCD's model, the construction of the LPP is financed by the State of Utah. Presumably the State gets the money by issuing bonds, but the District's model says nothing about this. (If the State gets the money from ongoing revenues instead of bonds, the State incurs an opportunity cost, the interest rate on the State's savings, because without the LPP to pay for, the State could invest the money—or spend it in an even more productive way, which would cause the opportunity cost of instead paying for the LPP to be even higher.) The District makes an initial payment to the State of \$50 million in 2014. When the District takes its first water delivery, which is in 2032 (Column W) and which is 1.2% of the pipeline's capacity (Row 23), it pays to the State 1.2% of "the LPP cost (\$969 million, 'LPP Cost'!\$I\$17) minus the initial \$50 million payment

---

<sup>1</sup>A minor spreadsheet correction, which changes conclusions by less than 2%, is at [http://economics.utah.edu/lozada/Research/2015\\_LPP\\_FINAL\\_Repayment\\_Analysis9.Windows.xlsm](http://economics.utah.edu/lozada/Research/2015_LPP_FINAL_Repayment_Analysis9.Windows.xlsm).

('Initial Capital Payment' ! \$C\$10).'' This is repeated in subsequent years: each year's increment in population, times its GPCD, yields that year's new water required, and that, expressed as a percent of the LPP's capacity, determines the amount of LPP construction cost which the District must pay the State in that year. This continues until 2061, when all the pipeline's water is being used and all the State's construction costs have been paid back by the District. The District never reimburses the State for the interest payments which the State made.

In the spreadsheet's base case, the District pays half of what it owes through water rates ('Incremental Repayments' I22) and the rest through impact fees (a one-time fee collected for each new connection to the water system). In this base case, the portion collected through impact fees is raised in full each year through the fees imposed that year ('Incremental Repayments' I25, "pay-go"). However, the portion collected through water rates is financed by the District issuing debt (Incremental Repayments' I22) for 30 years ('Incremental Repayments' I31) at an interest rate of 4.5% ('Interest Rate' ! \$J\$16). The District's debt is assumed to be fully-amortizing, so annual payments of interest *and principal* are due.

The following problems exist in the WCWCD's model. (Each set of problems is explained in more detail in the appendices.)

1. Errors in counting, especially counting the number of repayment periods for bonds (Appendix A).
2. Not reimbursing the State for the interest which the State has to pay on the bonds the State issues to pay for the project. The State has to pay interest (or, as stated above, opportunity costs) starting in 2014 and ending in {2031, 2032, 2033, . . . , 2061}, but this model assumes the District only repays the State's principal on those dates, not any of the State's interest.<sup>2</sup> Note that Utah Code 73-28-402 (4) (part of the Lake Powell Pipeline Development Act) states: "The board shall establish and charge a reasonable interest rate for the unpaid balance of reimbursable preconstruction and construction costs." See Appendices B and C. The former reports that, using the model's default parameters, omitting these interest costs implies that the WCWCD only repays 28% of the project's present discounted value.
3. Omission of "operations and maintenance" costs. See Appendices D and E.

These mistakes are all fixed in our LAFP (3), which is available at [http://economics.utah.edu/lozada/Research/Model\\_of\\_WCWCD\\_7.xlsx](http://economics.utah.edu/lozada/Research/Model_of_WCWCD_7.xlsx). Figure 1 compares per-capita costs through time when these corrections are made. (The Graph's "original WCWCD" line already corrects the mistakes in (1).) The graph goes to 2090 (rather than the 2060 end date of the WCWCD graphs and the analysis of the Economists' model) because, as explained in footnote 2, the District does not finish paying off its LPP bonds until then, even though modeling conditions long into the future is unlikely to be accurate.

<sup>2</sup>For 30 years after {2031, 2032, 2033, . . . , 2061}, the District does pay interest on *its own* bonds.

Figure 1 shows that the District has greatly underestimated the LPP’s cost burden on Washington County residents.

However, the WCWCD’s model is much more flawed than this. To see why, for the rest of this section assume that we *have already fixed* errors (1), (2), and (3) in the WCWCD model, so that when we refer to “the WCWCD model” from now on, it will be the one in our LAFP (3), not to the original one in LAFP.

Recall the structure of the WCWCD model: in each year, given the default parameters, population growth determines the total quantity of water used; that, in turn, determines the increase in water used compared to the previous year, and this increase determines the total revenue required from selling that water (total revenue required to pay off the District’s bonds). In other words, population determines the *total quantity of water*, and the *incremental quantity of water* determines the *total revenue*; but if there is a market demand curve for water then total quantity, once determined, will in turn determine total revenue, leaving no scope for anything else (such as the incremental water, or revenue needs) to determine total revenue. The WCWCD’s model addresses that problem by pretending that the market demand curve for water does not exist, but that is a fatal flaw.

Figure 2 illustrates this problem, taking the year 2050 as an example. Use  $Q$  to denote quantity of water and  $P$  to denote price of water. The WCWCD projects 2050’s  $Q = 35.15$  billion gallons, shown on the horizontal axis. The WCWCD projects 2015’s “total revenue”  $TR = \$86$  million, because this is how much money is needed from water rates (not from impact fees) to pay for the LPP debt that year. Since  $TR = P \times Q$ , these two figures imply a price  $P = TR/Q = \$86 \text{ million}/35.15 \text{ billion gallons} = \$2,446,657/\text{billion gallons}$ . The “X” in Figure 2 shows that combination of price and quantity. However, we can project a water demand curve for 2050, as in the prior report: take demand to be proportional to  $1/\sqrt{P}$  and assume demand grows over time with population. That gives rise to the demand curve illustrated in Figure 2, and it is clear that the “X” does not lie on this demand curve. So the WCWCD’s projections for  $TR$  and for  $Q$  are inconsistent. This is to be expected, since the WCWCD’s model does not acknowledge that demand curves exist.

To extend this example to other dates, let  $\alpha$  to denote a constant, use  $\beta$  to denote one plus the projected rate of population growth, and use subscripts to denote time periods. Then

$$\begin{aligned} Q_{2014} &= \alpha P_{2014}^{-1/2} = \beta^0 \alpha P_{2014}^{-1/2} \\ Q_{2015} &= \beta^1 \alpha P_{2015}^{-1/2} \\ Q_{2016} &= \beta^2 \alpha P_{2016}^{-1/2} \\ Q_{2017} &= \beta^3 \alpha P_{2017}^{-1/2} \dots \end{aligned}$$

In our worksheet ‘LPP Act Financing Plan (3)’, since 2014’s  $Q$  is E10 and 2014’s  $TR$  is C150 (which is 2013’s total revenue times 2.9%, which is ‘Population Inp ut’ I15, “annual population growth 2009–2060), one can obtain  $\alpha$ , and its value is in Cell C151.




Figure 1. Cost per person per year (in current dollars, not in 2014 dollars), WCWCD default parameters (\$969 million project cost, 4.5% interest rate, 50/50 division between water rates and impact fees). For the interpretation of the vertical dashed lines, see Figure 3. For data sources, see Appendix F.

Solid line: according to the WCWCD model.

Dotted line: WCWCD model with added interest payments reimbursing the State for the State's interest payments.

Dashed line: WCWCD model with added interest payments and operations and maintenance expenses for the LPP.

Dashed-dotted line: WCWCD model with added interest payments and operations and maintenance expenses for LPP and for the currently-existing infrastructure.




Figure 2. Water's price  $P$ , quantity  $Q$ , and demand in 2050. The "X" marks the spot where the WCWCD's projections for  $Q$  and for  $TR$  are consistent with each other. However, this point is not on the demand curve. A  $Q$  of 35.15 billion gallons will not lead to a  $TR$  of \$86 million, but only to a  $TR$  of \$24 million. To get  $TR = \$86$  million requires a very different combination of price and quantity, shown by the shaded rectangle.




Figure 3. Solid line: the “total revenue” in the WCWCD model which is consistent with the quantity of water in the WCWCD model and with our projected demand curves. Dashed line: the WCWCD’s projection of total revenue, determined, as in that model, by the need to repay LPP costs. Between the vertical dashed lines (which are the same in Figure 1), the WCWCD’s revenue projections were too high.

One has

$$\begin{aligned}
 TR_{2014} &= P_{2014}Q_{2014} = \alpha P_{2014}^{1/2} = \alpha^2/Q_{2014} \\
 TR_{2015} &= P_{2015}Q_{2015} = \beta\alpha P_{2015}^{1/2} = \beta\alpha \frac{\beta\alpha}{Q_{2015}} = \beta^2\alpha^2/Q_{2015} \\
 TR_{2016} &= P_{2016}Q_{2016} = \beta^2\alpha P_{2016}^{1/2} = \beta^2\alpha \frac{\beta^2\alpha}{Q_{2016}} = \beta^4\alpha^2/Q_{2016} \\
 TR_{2017} &= P_{2017}Q_{2017} = \beta^3\alpha P_{2017}^{1/2} = \beta^3\alpha \frac{\beta^3\alpha}{Q_{2017}} = \beta^6\alpha^2/Q_{2017} \dots
 \end{aligned}$$

To use this to obtain Figure 3, in Row 152 of 'LPP Act Financing Plan (3)' we assume the WCWCD’s water demand projection is correct. Then Cell I152, the correct total revenue from water sales for 2018, is

$$(1 + \text{'Dynamic Assumptions'!G\$2})^{2 * (I4 - \$E\$4)} * \$C\$151^2 / (I10 * 1000000)$$

that is,

$$\beta^{2 * (2018 - 2014)} * \alpha^2 / Q_{2018} .$$

Figure 3 illustrates Row 152, the correct total revenue from water sales (that is, the total revenue that is consistent with the assumed water sales, assuming our water demand curve). For contrast, it also shows the total revenue from water sales which WCWCD projected (Row 128). If the WCWCD’s model had been consistent with demand curves, the two lines in Figure 3 would have been identical.

It is not possible to fix this. The model cannot be rescued because it assumes quantity can be driven by population growth while total revenue is driven by a percentage of project cost utilized, and those assumptions are inconsistent, not just with our a market demand curve, but with any market demand curve (except one artificially and completely

implausibly contrived to shift in such a way that it fits at each date), because a market demand curve implies that once quantity is determined, total revenue is also determined; in other words, total revenue is determined by total quantity, so total revenue cannot simultaneously be determined by LPP financing needs, or by the incremental increase in water use over time.

## **Section II. The Model endorsed by the Economists**

We will call this “the economists’ model.”

The ‘Population Estimates’ sheet comes from the GOPB 2012 projections.

The DSWRESR sheet is data taken from the “Draft Socioeconomics and Water Resource Economics Study Report.”

The Revenues and Expenses sheet is further data taken from the District or the DSWRESR.

The First Scenario and Second Scenario sheet are identical, except one can choose parameters such as B6 (interest rate on reserves), N6 (choice of low-cost or high-cost project—the high-cost has more “pumped storage” electrical generating capacity), N10 (initial payment-free years), and N11 (interest rate on debts).

Column B are Property Taxes, assumed to grow at the “household growth rate” B5 every year.

Column C is water sales revenue, tentatively assumed to grow at B5 yearly, as it would if price were constant at its historical level and demand grew at B5. (“Tentatively” is explained in Column Q.)

Column D is “power sale revenue and surcharges,” assumed to grow at B5 yearly.

Column E is tentative “impact fees.” (“Tentative” is explained in Column S.) For example, Cell E27 is

$$=B3*'Population Estimates'!E$4*B$5^(A27-$A$24)*B$4$$

B3 is the 2013 impact fee per ERU. ‘Population Estimates’!E\$4 is the number of households in 2012. B5 is the household growth rate plus one. (A27-\$A\$24) is the number of years beyond 2015. Combining these, ‘Population Estimates’!E\$4\*B\$5^(A27-\$A\$24) is the number of households in 2018. B4 is the household growth rate, so multiplied by the number of households in 2018 gives the number of new households in 2018.

Column F lists revenue anticipated from selling some WCWCD land.

Column G lists LPP electric power revenue.

Column H adds all the revenue (tentative and definite).

Column I lists annual debt service on WCWCD debt which pre-dates the LPP. Column J lists operations and maintenance costs for the WCWCD’s non-LPP existing infrastructure.

Column K gives annual LPP debt service, which is a constant every year, and comes from N14, =-PMT(N11,N12,N13), where N11 is the interest rate as above, N12 is the number of years allowed to pay off the loan, and N13 is the loan amount N7 inflated by negative amortization if there is an initial payment-free period.



Column L are LPP operations and maintenance costs.

Column M is the total annual debt service, Column K plus Column I. Column I was off by one year in this sum; 2015 LPP FINAL Repayment Analysis9.Windows.xlsm fixes this minor error, which affects the concluding numbers by approximately 1%.<sup>3</sup>

Column N, total expenses, sums Columns J, M, and L. Column O, net annual surplus, is total revenues (Column H) minus total expenses (Column N). Column P, cumulative surplus, adds the current year's net surplus (Column O) to B6 (which is one plus the interest rate on reserves) times the previous year's cumulative surplus.

Column Q is "Repayment Option 1: Annual Surplus (Deficit) w/ Increased Water Rate sale revenue." The code is, for Q26:

$$=IF(ROW()-24<N$10,026,026+$B$10*C26)$$

N10 is the number of initial payment-free years. O26 is the net annual surplus or deficit. C26 is tentative water sales revenue. B10 is "the factor by which water sales revenue needs to increase to eliminate the debt by 2062, minus one." B10 is the same as S75, and it is determined by an iterative numerical process that has the goal of making R73, the final-year cumulative surplus or deficit when water rates are the instrument, equal to zero; so the other entries of Column R are each year's cumulative surplus or deficit when water rates are the instrument and where water sales revenue is increased by a factor of B10 plus one.

Column S is "Repayment Option 2: Annual Surplus (Deficit) w/ Increased Impact Fees." The code is, for S26:

$$=IF(ROW()-24<N$10,026,026+$B$15*E26)$$

N10 is the number of initial payment-free years. O26 is the net annual surplus or deficit. E26 is tentative impact fee revenue. B15 is "the factor by which Impact Fees need to increase to eliminate the debt by 2062, minus one." B15 is the same as U75, and it is determined by an iterative numerical process that has the goal of making T73, the final-year cumulative surplus or deficit when impact fees are the instrument, equal to zero; so the other entries of Column T are each year's cumulative surplus or deficit when impact fees are the instrument and where impact fee revenue is increased by a factor of B15 plus one.

Columns U and V use a 50/50 split between Impact Fees and Water Rates, and are a linear combination of Columns Q and S, on the one hand, and R and T, on the other. However, the general formula for U26 is

$$=IF(ROW()-24<N$10,026,026 + $M$19*C26*$B$10 + $M$18*E26*$B$15)$$

<sup>3</sup>See footnote 1. As examples of how this changes the conclusions, here are the new (first) and old (second) figures from two sentences in the Economists' 2015 letter. "Assuming the \$1.8 billion high-cost LPP alternative from the 2012 Socioeconomics and Water Resource Economics Report, the District could raise the needed funds by: raising impact fees 137/138 percent, to an average of \$14,473/\$14,514 per connection; together with raising water rates by 673/678 percent. . . ." "Assuming the \$1.4 billion low-cost LPP alternative from the 2012 Socioeconomics and Water Resource Economics Report, the District could raise the needed funds by: raising impact fees 123/123 percent, to an average of \$13,598/\$13,630 per connection; together with raising water rates by 573/576 percent. . . ."

where M18 is “impact fee’s portion of split financing” and M19 is one minus M18. The general formula for V26 is

$$=V25*\$B\$6+U26$$

analogously with Columns P, R, and T.

Once the change in water sales revenue has been determined, the previous report explained how to derive the required change in water prices and in water demand, assuming the demand curve for water is isoelastic with an elasticity of  $-0.5$ .

Impact fees obviously do not change water demand, but, to the extent that the supply curve for land is vertical, impact fees reduce the price of land on a one-for-one basis. Hence the incidence of impact fees falls 100% on existing landowners, who may think this figure will be 0% (i.e., that newcomers will pay the impact fee, which is how it looks superficially).

### Appendix A. Initial Corrections to the WCWCD’s Analysis

#### 1. LAFP rows 54–82: debt repayment, “Series 2032A Bonds.”

Consider Row 54. The row has no numbers until Column W; cell W3 identifies this as “Year 19,” and cell W4 says 2032. So Column W is 2032. Row 54 has numbers from then until Column BA corresponding to 2062: 30 years. When bond traders refer to “Series 2032 bonds,” they mean that 2032 is the year the bonds are issued. So what’s happening is that the State of Utah issues “series 2014” bonds. Say it’s for \$100 just for illustration. The State of Utah takes care of all the interest payments until 2032. Standard municipal bonds are interest-only bonds (they don’t amortize), so in 2032, the State of Utah all of a sudden has to pay back \$100. The District’s understanding of “pay as you go” is: in 2032, the District needs to pay the State \$100 (so the State can pay its bondholders \$100). But in 2032 the District doesn’t have on hand the \$100. So in 2032 the District borrows \$100. It then pays this debt back with interest (using amortization, the spreadsheet assumes) until it’s paid up in 2062. Upshot: the State has given the District an interest-free loan of \$100 until 2032.

Cell W54 (year 2031), money due in 2031 for bonds issued in that year:

```
=IF('Incremental Repayments'!$R$22="Debt",
    IF(AND(W$4>=$D54,W$4<=$D54+'Incremental Repayments'!$I$31),
        -PMT('Interest Rate'!$J$16,
            'Incremental Repayments'!$I$31,
            (HLOOKUP($D54,$E$4:$CZ$32,28,FALSE))*
            'Incremental Repayments'!$I$22))
    ,0)
,0)
```

Cell W54 (year 2031), with inserted explanations:

```
=IF('Incremental Repayments'!$R$22="Debt", <always true>
    IF(AND(W$4>=$D54,W$4<=$D54+'Incremental Repayments'!$I$31),
```

W\$4 is 2032.

\$D54 is 2032; \$D55 will be 2033; \$D56 will be 2034....

'Incremental Repayments'!\$I\$31 is 30 years.  
 \$D54+'Incremental Repayments'!\$I\$31 is 2062, so payments will occur for  
 the years 2032, 2033, ..., 2062, which is 31 years.

```
-PMT('Interest Rate'!$J$16,  

'Incremental Repayments'!$I$31,
```

The second argument of PMT is the number of payments, here 30; so the annual payment is the one appropriate for 30 years, but from above, there will be 31 years of payments (a mistake).

```
(HLOOKUP($D54,$E$4:$CZ$32,28,FALSE)*
```

Look for the value of \$D54 (which is 2032) in the top row of the table \$E\$4:\$CZ\$32 (the numerical entries of the 4th through 32nd rows of this spreadsheet); then return the value in the 28th row of that column, which is Spreadsheet Row 28+4-1=31, "additional capital payments," which is what the State is owed in year 2032. So this is the loan principal.

```
'Incremental Repayments'!$I$22)
```

(the share allocated to water rates instead of impact fees)

```
) <end of PMT>  

,0) <end of 2nd IF; return 0 if the year is out-of-range>  

,0) <end of 1st IF>
```

The fix: use

```
=IF('Incremental Repayments'!$R$22="Debt",  

IF(AND(W$4>=$D54,W$4<$D54+'Incremental Repayments'!$I$31),  

< note this strict inequality  

-PMT('Interest Rate'!$J$16,  

'Incremental Repayments'!$I$31,  

(HLOOKUP($D54,$E$4:$CZ$32,28,FALSE)*  

'Incremental Repayments'!$I$22))  

,0)  

,0)
```

instead of

```
=IF('Incremental Repayments'!$R$22="Debt",  

IF(AND(W$4>=$D54,W$4<=$D54+'Incremental Repayments'!$I$31),  

<= note this weak inequality  

-PMT('Interest Rate'!$J$16,  

'Incremental Repayments'!$I$31,  

(HLOOKUP($D54,$E$4:$CZ$32,28,FALSE)*  

'Incremental Repayments'!$I$22))  

,0)  

,0)
```

The analogous fixes have to be made for all cells in the rectangle defined by E54:CZ123.

The spreadsheet assumes the State gives the District one interest-free loan from 2014 until 2033. The State gives the District another, partially-overlapping interest-free loan from 2014 until 2034. And another one from 2014 until 2035. And so on, until the last interest-free loan stretches from 2014 until 2060.

In Row 53, the District is making some payments starting in 2014, so they don't get interest-free use of 100% of the LPP's costs. They are making payments starting in 2014 on the \$50 million "initial capital payment." But that \$50 million is the only money they are paying the full cost of. All their other LPP financing is being subsidized for between two and five decades by the State.

2. *LAFP row 53: debt repayment, "Series 2025 A Bonds, Initial Capital Payment."*

These extend from Column E (year 2014) to Column AT (year 2055), a total of  $2055 - 2014 + 1 = 42$  payments. Since they start in 2014, they contain no interest rate subsidy from the State.

Cell E53 is:

```
=IF(AND('Initial Capital Payment'!$C$14="Debt", <the default>
E4<=('Economist Prepay (Plan 2)'!$I$16+
'Initial Capital Payment'!$F$7),
```

E4 is 2014.

'Economist Prepay (Plan 2)'!\$I\$16 is 2025, "in what year will WCWCD make the payment" ("prepayment of construction costs").

'Initial Capital Payment'!\$F\$7 is 30 years.

```
E4>='Initial Capital Payment'!$C$12), <2014, the year of payment>
```

So this code checks to make sure the year (E4, F4, ...) is between 2014 and  $2025 + 30 = 2055$  (Column AT), which is 42 years.

```
-PMT('Interest Rate'!$J$16,
'Initial Capital Payment'!$F$7, <30 years, mismatching 42
years>
'Initial Capital Payment'!$C$10) <$50 million initial
capital payment>
,0) <end of beginning IF>
```

The mistake comes in using two inconsistent values for the date when the initial capital payment occurs, 'Economist Prepay (Plan 2)'!\$I\$16 and 'Initial Capital Payment'!\$C\$12. The fix is to use just one of these; I chose the second one:

```
=IF(AND('Initial Capital Payment'!$C$14="Debt",
E4<('Initial Capital Payment'!$C$12+'Initial Capital Payment'!$F$7),
E4>='Initial Capital Payment'!$C$12),
```

In line 2 note < replaces <=

```
-PMT('Interest Rate'!$J$16,
'Initial Capital Payment'!$F$7,
'Initial Capital Payment'!$C$10),0)
```

3. *“Incremental Share of Lake Powell Capacity Utilized, Annual”*

In the WCWCD’s Line 23 of LAFP, in Year 47 (that is, 2060), 97.7% of the pipeline is utilized. The next year, 100% is utilized. So in Year 48, the incremental share utilized should be  $100\% - 97.7\% = 2.3\%$ . In the WCWCD’s spreadsheet, it is zero. This is fixed (analogously for the whole row, although this is the only cell which changes) by replacing

```
=IF(AZ24<100%,(AZ21-AY21)*1000000/('Construction Timing'!$I$23*325851),0)
```

with

```
=IF(AZ24<100%,(AZ21-AY21)*1000000/('Construction Timing'!$I$23*325851), IF(AND(AZ24=100%, AY24<100%), AZ24-AY24,0))
```

These counting errors are fixed in our version of LAFP, abbreviated from now on LPPAFP.

4. *Population, 2081–2090*

In order to populate these cells (BT6–CC6 of LPPAFP), we filled in “Economist Plan 1” BR5 to CB5 (years 2080 to 2090) and BR7 to CB7 (population). In turn, these populated columns BT to CC (years 2081 to 2090) of Row 134 (Impact Assessment’s per capita annual cost) and Row 135 (Impact Assessment’s “cost per household per month”).

5. *Water Demand, 2081–2090*

In order to fill this in (BT9–CC9), we had to fill in “Economist Plan 2” BS9–CB9 (water demand, gallons per capita per day—GPCD), which required filling in “Economist Plan 1” BR9–CB9, which we did.

This, in turn, filled in LPPAFP Row 133 columns BT–CC.

6. *Impact Fees, 2081–2090*

In order to fill in LPPAFP Row 136 columns BT–CC, one has to fill in BT7–CC7, which requires filling in “Economist Plan 2” BS8–CB8, which requires filling in “Economist Plan 1” BR8–CB8 (“Incremental ERU’s). Which we did.

**Appendix B. Incorporating Interest: Diagnosing the Problem in LPPAFP**

1. New Row 52 gives the discount factors.
2. New Row 125 discounts Row 124, “Total Bond Repayment,” by the discount factors.
3. New Row 131 discounts Row 130, “Cumulative/Combined Cost to Washington County Taxpayers,” by the discount factors.
4. New Cell 138 reports the final value of Row 131, and New Cell E139 reports that after payments are discounted, what “percent of project [is] paid for by [the] WCWCD.” With the WCWCD’s default parameters, it is 28%.

**Appendix C. Incorporating Interest: Fixing the Problem in LAFP (2)**

LPPAFP was copied onto LAFP (2), then the following changes were made there.

1. The generic entry in Row 27, “Beginning Capital Balance,” changed from (using AB27 as an example)

$$=AA32$$

(which is the previous year’s “Ending Capital Balance”) to

$$=AA32*(1+'Interest Rate'!$J$16)$$

2. The generic entry in Row 31, “Additional Capital Payments [due],” was (using W31 as an example)

$$=W23*( 'LPP Cost'!$I$17-'Initial Capital Payment'!$C$10)$$

which is “Incremental Share of LPP Capacity Utilized” times the “initial project cost minus the initial capital payment”—note that the initial project cost is not adjusted for interest. The sum of these “additional capital payments” is

$$\begin{aligned} &W23*( 'LPP Cost'!$I$17-'Initial Capital Payment'!$C$10) + \\ &X23*( 'LPP Cost'!$I$17-'Initial Capital Payment'!$C$10) + \\ &Y23*( 'LPP Cost'!$I$17-'Initial Capital Payment'!$C$10) + \dots \\ &= \\ &1 *( 'LPP Cost'!$I$17-'Initial Capital Payment'!$C$10) \end{aligned}$$

because the sum of the incremental LPP shares is one. So the LPP is finally paid off in full, but without interest.

In our model, W31 (additional capital payments) is changed to

$$=W24*W27$$

which is “Cumulative share of LPP capacity utilized” times “Beginning Capital Balance,” where the “Beginning Capital Balance” shrinks because of the previous year’s payment (unlike ( 'LPP Cost'!\$I\$17-'Initial Capital Payment'!\$C\$10) which is a constant) but also rises because of the accrual of interest. This is the correct reformulation because it guarantees that once 100% of the water from the LPP is taken by the District, 100% of the project costs must have been paid to the State. If instead annual payments were “Incremental share of LPP capacity utilized” times “Beginning Capital Balance,” then since “Incremental share of LPP capacity utilized” is always strictly less than one, in no year would the “Beginning Capital Balance” ever be completely paid off.

**Appendix D. Adding LPP’s O&M Costs to LAFP (2)**

New Row 139 uses the FERC “low-cost” estimate for Operations & Maintenance Costs in Cell W139, then increases it at 4%/year in X139–CZ139.

New Row 140 lists Row 139’s costs in 2014 dollars, using the overall interest rate of the spreadsheet, 'Interest Rate'!\$J\$16.

New Row 141 gives “Combined Cost per capita, annual, including LPP O&M.”

### **Appendix E. Adding LPP's O&M Costs, in LAFP (3)**

LAFP (2) was copied onto LAFP (3), then the following changes were made there.

New Row 140 uses District's estimate for Operations & Maintenance Costs for its existing infrastructure for 2015 in Cell F140. Then the next cell, G140, is  $F140 * (1 + \text{'Dynamic Assumptions'!G2})$  where the last term is equal to the assumed annual population growth rate, "Population Input" I15.

New Row 141 lists Row 140's costs in 2014 dollars, using the overall interest rate of the spreadsheet, 'Interest Rate'!J\$16.

New Row 142 gives "Combined Cost per capita, annual, including both O&M."

### **Appendix F. Sources for Figure 1**

Original WCWCD:

= 'LPP Act Financing Plan'!\$F\$137:\$CC\$137

Original WCWCD plus Interest:

= 'LPP Act Financing Plan (2)'!\$F\$132:\$CC\$132

Original WCWCD plus Interest plus O&M for LPP:

= 'LPP Act Financing Plan (2)'!\$F\$141:\$CC\$141

Original WCWCD plus Interest plus O&M for LPP and Existing:

= 'LPP Act Financing Plan (3)'!\$F\$142:\$CC\$142